Work done pushing a book along a table

Suppose I push a book along a table at an angle of 30° *below* the horizontal. Starting from rest, I push the book a distance of 2 m in a time of 8 seconds. The book has mass 5 kg, and there is a coefficient of kinetic friction 0.3 between the book and the table. What is the total work done by me? How much work is done by all the other forces in the problem?

Work done pushing a book along a table

Suppose I push a book along a table at an angle of 30° *below* the horizontal. Starting from rest, I push the book a distance of 2 m in a time of 8 seconds. The book has mass 5 kg, and there is a coefficient of kinetic friction 0.3 between the book and the table.

In[ ]:=
```
Clear[d, t, v0, vf, a, m, g, μk, Fme, θ, Fn, fk, Wme, Wg, Wn, Wfr, Wnet, Ki, Kf]
```

What is the work done by me?

In[ ]:=
```
Wme = Fme * d * Cos[θ]
```

Out[ ]=
```
d Fme Cos[θ]
```

We need to figure out the force. Use Newton's 2nd law ΣF = ma. First, find the acceleration, and then

break Newton's second law into components.

What is the acceleration?

```
In[ ]:= d = Quantity[2.0, "Meters"]; t = Quantity[8.0, "Seconds"];
        v0 = Quantity[0, "Meters" / "Seconds"];
        a = a /. Solve[d == v0 t + (1 / 2) a t^2, a][[1]]
```

```
Out[ ]= 0.0625 m/s^2
```

What is the Force?

X-components:

```
In[ ]:= xeq = Fme Cos[θ] - fk == m a
```

```
Out[ ]= -fk + Fme Cos[θ] == m (0.0625 m/s^2)
```

```
In[ ]:= fk = μk Fn;
```

Y - Components

```
In[ ]:= yeq = Fn - m g + Fme Sin[θ] == 0
```

```
Out[ ]= Fn - g m + Fme Sin[θ] == 0
```

```
In[ ]:= m = Quantity[5.0, "Kilograms"]; g = Quantity[9.8, "Meters" / "Seconds"^2];
        μk = 0.3; θ = -30°;
```

```
In[ ]:= Solve[{xeq, yeq}, {Fn, Fme}]
```

```
Out[ ]= {{Fn → 59.4832 N, Fme → 20.9664 N}}
```

```
In[ ]:= {Fn, Fme} = {Fn, Fme} /. %[[1]]
```

```
Out[ ]= {59.4832 N, 20.9664 N}
```

Work done by various forces:

```
In[ ]:= Wme = UnitConvert[Fme d Cos[θ], "Joules"]
```

```
Out[ ]= 36.3149 J
```

```
In[ ]:= Wg = m g d Cos[90°]
```

```
Out[ ]= 0. kg m^2/s^2
```

```
In[ ]:= Wn = Fn d Cos[90°]
```

```
Out[ ]= 0. m N
```

In[ ]:=
```
Wfr = UnitConvert[μk Fn d Cos[180 °], "Joules"]
```

Out[ ]=
```
-35.6899 J
```

In[ ]:=
```
Wnet = Wme + Wg + Wn + Wfr
```

Out[ ]=
```
0.625 J
```

Kinetic Energies

In[ ]:=
```
Ki = (1 / 2) m v0^2
```

Out[ ]=
```
0. kg m^2/s^2
```

In[ ]:=
```
Clear[vf]
vf = vf /. Solve[vf^2 == v0^2 + 2 a d && vf > 0, vf][[1]]
```

Out[ ]=
```
0.5 m/s
```

In[ ]:=
```
Kf = UnitConvert[(1 / 2) m vf^2, "Joules"]
```

Out[ ]=
```
0.625 J
```